# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Contributor: Stefano Esposito <ragnarok@email.it>

pkgname=emacs-php-mode
pkgver=1.24.3
pkgrel=1
license=("GPL")
pkgdesc="PHP mode for emacs"
url="https://github.com/ejmr/php-mode"
arch=('any')
depends=('emacs')
options=(docs zipman)
install=emacs-php-mode.install
source=("$pkgname-$pkgver.tar.gz::https://github.com/ejmr/php-mode/archive/v$pkgver.tar.gz")
sha256sums=('22d8c62cd1718e041beac23b8affe410f20105630409486c2565b7210c694c0d')

build() {
  cd "$srcdir"/php-mode-$pkgver
  make
}

package() {
  cd "$srcdir"/php-mode-$pkgver/lisp
  for i in *.el{,c}; do
    install -Dm0644 $i "$pkgdir"/usr/share/emacs/site-lisp/php-mode/$i
  done
}
